Open-File Report

# RELA: REgional Liquefaction Assessment, An Interactive Computer Program

Albert T. F. Chen

U.S. Geological Survey
Menlo Park, California 94025

Open-File Report 85-468

This report is preliminary and has not been reviewed for conformity with U.S. Geological Survey editorial standards. Any use of trade names is for descriptive purposes only and does not imply endorsement by the USGS.

May, 1985

## INTRODUCTION

Seed and others (1982, 1983) have developed a simple yet effective procedure through which liquefaction potential of sand deposits can be determined on the basis of field testing and small amount of laboratory testing. The approach, known as the simplified procedure, uses standard penetration test (SPT) or cone penetration test (CPT) data as input and is applicable to sand and silty sand deposits.

An earlier report described the computer program PETAL which performs various steps of computations specified in the simplified procedure and assesses the liquefaction potential in terms of cyclic stress ratio, $\tau_{av}/\sigma'_v$, and modified penetration resistance, $N_1$ (Chen, 1984). For certain types of investigation, it may be more effective to evaluate the liquefaction potential in terms of depth and uncorrected penetration resistance, especially if the investigation is concerned with similar deposits in a given region or with changes in groundwater conditions.

The report describes the computer program RELA (for REgional Liquefaction Assessment) which generates liqufaction potential boundary curves in terms of depth and uncorrected penetration resistance for specified groundwater conditions. RELA is coded in FORTRAN and programmed to run in an interactive mode with a VAX 11/780 computer. The storage requirement to execute RELA is less than 15-K bytes and thus the program can be easily modified to run on many personal computers.

## GENERAL DESCRIPTION

Typical results from the simplified procedure or PETAL are shown in Fig. 1. The curve is the boundary between conditions of liquefaction and no liquefaction. If a point (A in Fig. 1, for example) is to the left of the boundary curve, the deposit whose estimated values of $\tau_{av}/\sigma'_v$ and $N_1$ correspond to those of that point is considered prone to liquefy. If a point (B for example) lies to the right of the boundary curve, the corresponding deposit is considered safe from liquefaction. The task that RELA performs is to generate the boundary curve in terms of depth and penetration resistance for different groundwater conditions such as those shown in Fig. 2, and therefore is essentially an inverse operation of PETAL. The criteria on which RELA is based are the same as those specified by the simplified procedure and used in PETAL. Specifics not included in this presentation are referred to the earlier reports.

For each depth, z, considered in RELA, subroutine STRESS first computes the overburden effective stress, $\sigma'_v$, and the total stress, $\sigma_v$, for the design groundwater condition and $\sigma'_v$ for the test groundwater condition. The design groundwater condition specifies the depth to groundwater table expected during the design earthquake and the test groundwater condition is referred to as the actual depth to groundwater table when penetration resistance was measured. RELA, however, does not consider the existence of capillary zones and, therefore, is not suitable for application to soil deposits in which the capillary zone causes significant changes in the effective stress. The stress quantities at both groundwater conditions are needed in converting the modified penetration resistance, $N_1$, to uncorrected penetration resistance as measured. From the $\sigma'_v$ and $\sigma_v$ for the design groundwater condition, the




Figure 1.-- Typical result from program PETAL




Figure 2.-- Typical result from program RELA

depth, and the peak acceleration specified for the design earthquake, the cyclic stress ratio, $\tau_{av}/\sigma'_v$ is estimated in the same manner as prescribed in the simplified procedure. If $\sigma'_v$ exceeds 1.5 ton/ft$^2$, $\tau_{av}/\sigma'_v$ is corrected to allow for the stress ratio reduction due to increasing confining pressure (Seed, 1983). A revised $N_1$ corresponding to the reduced stress ratio is then used for the conversion.

If the deposit in question qualifies as a silty sand, $N_1$ is first compensated for the correction made for grain size effects. A correction factor, $C_n$, is then determined according to the relative density of the deposit and the $\sigma'_v$ for the test groundwater condition. Dividing $N_1$ by $C_n$ results in the uncorrected penetration resistance. If applicable, this uncorrected penetration resistance is further compensated for the effect of shallow depth (less than 10 ft). In RELA, the uncorrected penetration resistance is expressed in terms of both the standard penetration test (SPT) in blow count/ft and the cone penetration test (CPT) tip resistance in Kg/cm$^2$.

A new feature in RELA is the consideration of the liquefaction potential of gravelly soils. This consideration is based on the fact that during cyclic shear, it takes more number of cycles for gravelly soils to reach peak pore pressure than for sands. If the ratio of time to reach peak pore pressure for a gravelly deposit to that for sands is known or can be assumed, a seperate relationship between normalized shear stress and number of cycles to cause liquefaction can be established. As illustrated in Fig. 3, the dotted curve for a gravelly sand is based on the results by Liu and others (1979) who found that for that particular gravelly sand of 45% gravel content, the number of cycles to reach peak pore pressure is 4 times of that for sands with zero




Figure 3.-- Relations of shear stress versus number of cycles to liquefy for different soils ($\tau_s$ is the shear stress to cause sand to liquefy in one cycle)

gravel content. RELA is capable of taking this type of relationships into account and generates the corresponding boundary curves accordingly.

As the program is coded in an interactive mode, all input to RELA are prompted and entered from the keyboard of remote computer terminals. The input required to execute RELA are self-explanatory as will be demonstrated later in the sample run. After the input are entered, RELA proceeds to perform computation for each depth considered and stores the results in a data file attached to I/O UNIT 16. The depths considered are generated internally from the depth to the design groundwater table downward to the cutoff depth acording to the input depth increment, dz. The cutoff depth is set to be the least of (1) the total depth of the deposit, (2) 80 ft, or (3) the depth at which the uncorrected penetration blow-count for the boundary curve exceeds 80 blow counts/ft. The latter two criteria are used because, as depth increases, estimated of $\tau_{av}/\sigma'_v$ based on the simplified procedure becomes less reliable and because correlation between high SPT blow-counts and soil behavior is not as well-established.

## SAMPLE RUN

For a demonstration run, consider a site consisting of two layers with their depths and densities listed below:

| | | |
|---|---|---|
| Layer 1 | 30 ft | 110 pcf |
| Layer 2 | 85 ft | 125 pcf |

The design groundwater condition is 15 ft below the surface, and the test groundwater condition is 50 ft. The design earthquake is assumed to have a 6.5 magnitude and a peak acceleration of 0.3 g. The depth increment, dz, is 3 ft.

Two sets of computations are to be performed. The first set considers the deposits as sands with a relative density of 55%. The second set treats the deposits as gravelly sands of the same density. The time to reach peak pore pressure for these gravelly sands is 3.5 times of that for sands.

The entire interactive session is reproduced and shown next. For distinction, input from the keyboard are printed in light italic. The output file from I/O UNIT 16 resulted from this sample run is also included. Data in columns 2, 9, and 10 are used for constructing the boundary curves.

(REPRODUCTION OF THE INTERACTIVE SESSION FOR THE SAMPLE RUN)

```
RELA: basic units are in LBS and FT

 enter title of this run in 72 characters or less
 demo run, set 1

 site description: enter no. of layers (<10)
 2
enter depth(ft) and density(pcf) with decimals of layer  1
 30.0,  110.0
enter depth(ft) and density(pcf) with decimals of layer  2
 85.0,  125.0

 enter expected depth of ground water during
 the design earthquake, and ground water depth
 when penetration test was performed -- 7.0, 20.0
 15.0,  50.0

 enter equake mag. and max acc (g) -- 7.5, 0.25
 6.5,  0.3

 enter depth increment, dz (1.0 to 5. ft)
 and relative density (0.4 for 40%), with a MINUS
 sign if sand is silty
 3.0,  0.55

 enter 0 if deposit is not gravelly
 0

 enter integer>0 for a new set of computation
 9


RELA: basic units are in LBS or FT

 enter title of this run in 72 characters or less
 demo run, set 2

 site description: enter no. of layers (<10)
 2
enter depth(ft) and density(pcf) with decimals of layer  1
 30.0,  110.0
enter depth(ft) and density(pcf) with decimals of layer  2
 85.0,  125.0

 enter expected depth of ground water during
 the design earthquake, and ground water depth
 when penetration test was performed -- 7.0, 20.0
 15.0,  50.0
```

```
enter equake mag. and max acc (g) -- 7.5, 0.25
6.5, 0.3

enter depth increment, dz (1.0 to 5. ft)
and relative density (0.4 for 40%), with a MINUS
sign if sand is silty
3.0, 0.55

enter 0 if deposit is not gravelly
9
enter multiple of time required for gravelly soil
to reach peak pore pressure when compared to
that for sand -- 4.1
3.5

enter integer>0 for a new set of computation
0
```

(OUTPUT FILE FROM UNIT 16)

```
demo run, set 1

the site consists of  2 layers w/ depths & dens:
                      1      30.0 (ft)          110.0 (pcf)
                      2      85.0 (ft)          125.0 (pcf)

input relative density =  0.55

input eq. mag.= 6.50   max. acc. = 0.30 g
design ground water table depth =  15.0 ft.
testing ground water table depth =  50.0 ft.
```

| count | depth | design stress (psf) effective | design stress (psf) total | testing stress (psf) effective | testing stress (psf) total | stress ratio | modified N1 | uncorrected spt-bc,N | cpt-Qc kg/cm2 | remarks |
|---|---|---|---|---|---|---|---|---|---|---|
| (1) | (2) | (3) | (4) | (5) | (6) | (7) | (8) | (9) | (10) | (11) |
| 1 | 15.0 | 1650.0 | 1650.0 | 1650.0 | 1650.0 | 0.19 | 15.4 | 13.9 | 62.7 | |
| 2 | 18.0 | 1792.8 | 1980.0 | 1980.0 | 1980.0 | 0.21 | 16.8 | 16.6 | 74.6 | |
| 3 | 21.0 | 1935.6 | 2310.0 | 2310.0 | 2310.0 | 0.22 | 17.9 | 19.1 | 85.8 | |
| 4 | 24.0 | 2078.4 | 2640.0 | 2640.0 | 2640.0 | 0.23 | 18.9 | 21.5 | 96.6 | |
| 5 | 27.0 | 2221.2 | 2970.0 | 2970.0 | 2970.0 | 0.24 | 19.7 | 23.8 | 106.9 | |
| 6 | 30.0 | 2364.0 | 3300.0 | 3300.0 | 3300.0 | 0.25 | 20.3 | 25.9 | 116.6 | |
| 7 | 33.0 | 2551.8 | 3675.0 | 3675.0 | 3675.0 | 0.26 | 20.7 | 28.1 | 126.4 | |
| 8 | 36.0 | 2739.6 | 4050.0 | 4050.0 | 4050.0 | 0.26 | 20.8 | 30.2 | 135.8 | |
| 9 | 39.0 | 2927.4 | 4425.0 | 4425.0 | 4425.0 | 0.26 | 20.8 | 31.8 | 143.2 | |
| 10 | 42.0 | 3115.2 | 4800.0 | 4800.0 | 4800.0 | 0.26 | 21.4 | 34.3 | 154.2 | **Kd |
| 11 | 45.0 | 3303.0 | 5175.0 | 5175.0 | 5175.0 | 0.25 | 21.3 | 35.7 | 160.6 | **Kd |
| 12 | 48.0 | 3490.8 | 5550.0 | 5550.0 | 5550.0 | 0.25 | 21.1 | 36.9 | 166.2 | **Kd |
| 13 | 51.0 | 3678.6 | 5925.0 | 5862.6 | 5925.0 | 0.25 | 20.9 | 37.9 | 170.5 | **Kd |
| 14 | 54.0 | 3866.4 | 6300.0 | 6050.4 | 6300.0 | 0.24 | 20.7 | 38.2 | 171.8 | **Kd |
| 15 | 57.0 | 4054.2 | 6675.0 | 6238.2 | 6675.0 | 0.24 | 20.4 | 38.4 | 172.9 | **Kd |
| 16 | 60.0 | 4242.0 | 7050.0 | 6426.0 | 7050.0 | 0.23 | 20.1 | 38.6 | 173.6 | **Kd |
| 17 | 63.0 | 4429.8 | 7425.0 | 6613.8 | 7425.0 | 0.22 | 19.7 | 38.7 | 174.2 | **Kd |
| 18 | 66.0 | 4617.6 | 7800.0 | 6801.6 | 7800.0 | 0.22 | 19.4 | 38.7 | 174.2 | **Kd |
| 19 | 69.0 | 4805.4 | 8175.0 | 6989.4 | 8175.0 | 0.21 | 18.9 | 38.4 | 172.8 | **Kd |
| 20 | 72.0 | 4993.2 | 8550.0 | 7177.2 | 8550.0 | 0.21 | 18.5 | 38.0 | 171.2 | **Kd |
| 21 | 75.0 | 5181.0 | 8925.0 | 7365.0 | 8925.0 | 0.20 | 18.1 | 37.6 | 169.3 | **Kd |
| 22 | 78.0 | 5368.8 | 9300.0 | 7552.8 | 9300.0 | 0.19 | 17.6 | 37.1 | 167.2 | **Kd |

* NOTE: remark Cn implies effective overburden pressure out of range, Cn=1.8 is assumed
Kd implies that correction for eff. overburden pressure > 1.5 tons/sq.ft was applied.

* * * correction for gravelly sands included * * *

demo run, set 2

the site consists of 2 layers w/ depths & dens:

| | | |
|---|---|---|
| 1 | 30.0 (ft) | 110.0 (pcf) |
| 2 | 85.0 (ft) | 125.0 (pcf) |

input relative density = 0.55

input eq. mag.= 6.50 max. acc. = 0.30 g
design ground water table depth = 15.0 ft.
testing ground water table depth = 50.0 ft.

| count | depth | design stress (psf) effective | design stress (psf) total | testing stress (psf) effective | testing stress (psf) total | stress ratio | modified N1 | uncorrected spt-bc,N | cpt-Qc kg/cm2 | remarks |
|---|---|---|---|---|---|---|---|---|---|---|
| (1) | (2) | (3) | (4) | (5) | (6) | (7) | (8) | (9) | (10) | (11) |
| 1 | 15.0 | 1650.0 | 1650.0 | 1650.0 | 1650.0 | 0.19 | 10.9 | 9.9 | 44.6 | |
| 2 | 18.0 | 1792.8 | 1980.0 | 1980.0 | 1980.0 | 0.21 | 11.9 | 11.8 | 53.1 | |
| 3 | 21.0 | 1935.6 | 2310.0 | 2310.0 | 2310.0 | 0.22 | 12.8 | 13.6 | 61.1 | |
| 4 | 24.0 | 2078.4 | 2640.0 | 2640.0 | 2640.0 | 0.23 | 13.5 | 15.3 | 68.8 | |
| 5 | 27.0 | 2221.2 | 2970.0 | 2970.0 | 2970.0 | 0.24 | 14.0 | 16.9 | 76.1 | |
| 6 | 30.0 | 2364.0 | 3300.0 | 3300.0 | 3300.0 | 0.25 | 14.5 | 18.5 | 83.0 | |
| 7 | 33.0 | 2551.8 | 3675.0 | 3675.0 | 3675.0 | 0.26 | 14.7 | 20.0 | 90.0 | |
| 8 | 36.0 | 2739.6 | 4050.0 | 4050.0 | 4050.0 | 0.26 | 14.8 | 21.5 | 96.7 | |
| 9 | 39.0 | 2927.4 | 4425.0 | 4425.0 | 4425.0 | 0.26 | 14.8 | 22.7 | 102.0 | |
| 10 | 42.0 | 3115.2 | 4800.0 | 4800.0 | 4800.0 | 0.26 | 15.2 | 24.4 | 109.9 | **Kd |
| 11 | 45.0 | 3303.0 | 5175.0 | 5175.0 | 5175.0 | 0.25 | 15.2 | 25.4 | 114.4 | **Kd |
| 12 | 48.0 | 3490.8 | 5550.0 | 5550.0 | 5550.0 | 0.25 | 15.1 | 26.3 | 118.4 | **Kd |
| 13 | 51.0 | 3678.6 | 5925.0 | 5862.6 | 5925.0 | 0.25 | 14.9 | 27.0 | 121.4 | **Kd |
| 14 | 54.0 | 3866.4 | 6300.0 | 6050.4 | 6300.0 | 0.24 | 14.7 | 27.2 | 122.4 | **Kd |
| 15 | 57.0 | 4054.2 | 6675.0 | 6238.2 | 6675.0 | 0.24 | 14.5 | 27.4 | 123.1 | **Kd |
| 16 | 60.0 | 4242.0 | 7050.0 | 6426.0 | 7050.0 | 0.23 | 14.3 | 27.5 | 123.6 | **Kd |
| 17 | 63.0 | 4429.8 | 7425.0 | 6613.8 | 7425.0 | 0.22 | 14.1 | 27.6 | 124.0 | **Kd |
| 18 | 66.0 | 4617.6 | 7800.0 | 6801.6 | 7800.0 | 0.22 | 13.8 | 27.6 | 124.0 | **Kd |
| 19 | 69.0 | 4805.4 | 8175.0 | 6989.4 | 8175.0 | 0.21 | 13.5 | 27.3 | 123.0 | **Kd |
| 20 | 72.0 | 4993.2 | 8550.0 | 7177.2 | 8550.0 | 0.21 | 13.2 | 27.1 | 121.9 | **Kd |
| 21 | 75.0 | 5181.0 | 8925.0 | 7365.0 | 8925.0 | 0.20 | 12.9 | 26.8 | 120.5 | **Kd |
| 22 | 78.0 | 5368.8 | 9300.0 | 7552.8 | 9300.0 | 0.19 | 12.5 | 26.4 | 119.0 | **Kd |

* NOTE: remark Cn implies effective overburden pressure out of range, Cn=1.8 is assumed
Kd implies that correction for eff. overburden pressure $>$ 1.5 tons/sq.ft was applied.

## REFERENCES CITED


Chen, A. T. F., 1984, PETAL -- PEnetration Testing And Liquefaction, an interactive computer program: U.S. Geological Survey Open-File Report No. 84-290.

Liu, L., Li, K., and Bing, D., 1979, Earthquake damage of Baihe Dam and liquefaction characteristics of sand and gravel materials: Research Institute of Water Conservancy and Hydroelectric Power, Beijing, China.

Seed, H. B., 1983, Earthquake-resistant design of earth dams: Proceedings, ASCE Symposium on Seismic Design of Embankments and Caverns, American Society of Civil Engineers, New York, New York.

Seed, H. B., and Idriss, I. M., 1982, Ground Motion and Soil Liquefaction during Earthquakes: Engineering Monographs on Earthquake Criteria, Structural Design and Strong Motion Records, Earthquake Engineering Research Institute, Berkeley, California.

Seed, H. B., Idriss, I. M., and Arrango, I., 1983, Evaluation of liquefaction potential using field performance data: Journal of the Geotechnical Engineering Division, ASCE, Vol. 109, No. 3, March, pp. 458-482.


## Program Listing

If the user has access to the VAX 11/780 computer of the Office of Earthquakes, Volcanoes and Engineering, U.S. Geological Survey in Menlo Park, California, he can simply execute RELA by entering the command:

```
run pub1:[chen.liq]rela
```

and the output file for016.dat will be in the user's working directory.

Listing of RELA and its subroutines are reproduced in the following pages.

```
c         RELA: REgional Liquefaction Assessment
c
c               This program generates liquefaction potential curves
c               postulated in the simplified procedure in terms of depth
c               and uncorrected penetration resistance.  The program is
c               intended for the assessment of liquefaction potential of
c               similar deposits in a given region where considerations
c               of depth and ground water condition are important.
c
c         references: 1. Seed & Idriss, 1982, Ground Motions and Soil
c                        Liquefaction during Earthquakes: EERI Monograph
c                        Series.
c                     2. Chen, 1984, PETAL: USGS Open-File Report No.
c                        84-290.
c
c         by A. T. F. Chen, OEVE, U.S. Geological Survey, 6/84
c               revised 3/85 for documentation
c
c
c
      dimension dref(9), rd(9), sv8(16), cn8(16), sv4(16), cn4(16),
     & xf(20),yf(20), rmk(5),xn(9),yt(9),ut(9), title(18), resu(11,51)
      common /blka/den(9), th(9), depth(9), nlayer, zgw, zgwt
c
      data rmk/'    ',' cpt','silt',' *Cn','**Kd'/
c
c        digitized values of curve in fig. 40, ref. 1
c
      data rd/1.0,0.9794,0.9668,0.9478,0.9346,0.9189,0.9009,
     &        0.8709,0.40/
      data dref/0.0,11.825,15.469,21.643,27.268,31.752,34.813,
     &          39.535,100.0/
c
c        digitized values of the M=7.5 curve in fig. 57, ref. 1
c
      data xn/5.288,11.014,15.308,20.702,26.094,29.823,31.468,33.426,
     &        34.714/
      data ut/0.05333,0.1133,0.1588,0.2166,0.2765,0.3297,0.3529,0.3949,
     &        0.4379/
c
c        digitized values of curves in fig 47, ref. 1
c
      data sv8/0.7732,0.9447,1.2934,1.7221,1.9845,2.2949,2.6744,3.1689,
     &          3.5984,4.1400,4.7297,5.3664,6.1172,7.2153,8.1312,9.0241/
      data cn8/1.5965,1.4295,1.2288,1.0780,1.0114,0.9536,0.8951,0.8357,
     &          0.7952,0.7400,0.6936,0.6513,0.6035,0.5619,0.5310,0.5003/
      data sv4/0.7732,0.9447,1.2934,1.7221,1.9845,2.1597,2.5362,2.9828,
     &          3.4533,4.0370,4.5796,5.1473,5.8070,6.7640,7.7940,8.7560/
      data cn4/1.5965,1.4295,1.2288,1.0780,1.0114,0.9685,0.8963,0.8281,
     &          0.7643,0.6903,0.6397,0.5980,0.5556,0.5014,0.4649,0.4337/
c
c         format statements
c
```

```
    2 format(18a4)
    4 format('    ')
    6 format(' enter depth(ft) and density(pcf) with decimals of layer'
     & i3)
    8 format('1')
   16 format(/' the site consists of',i3 ' layers w/ depths &
dens:')
   18 format(20x,i4,f10.1,' (ft)',f15.1,' (pcf)')
   20 format(/' input eq. mag.=',f5.2,'   max. acc. =',f5.2,' g'/
     & ' design ground water table depth =  f6.1,' ft.'/
     & ' testing ground water table depth =',f6.1,' ft.'/)
   22 format(' count  depth  design stress (psf)  testing stress ',
     & '(psf)   stress   modified  uncorrected  cpt-Qc   remarks'/
     & 15x,'effective    total    effective    total     ratio',
     & '        N1       spt-bc,N     kg/cm2'/)
   23 format(3x,3h(1),4x,3h(2),6x,3h(3),8x 3h(4),7x,3h(5),8x,3h(6),
     & 7x,3h(7),8x,3h(8),8x,3h(9),8x,4h(10),5x,4h(11)/)
   24 format(i5,f7.1,2f10.1,2x 2f10.1 f10.2,f10.1,f11.1,f12.1,
     & 3x,a4,1x,a4)
   26 format(/' input relative density =',f6.2)
   28 format(//' * NOTE: remark Cn implies effective overburden',
     & ' pressure out of range, Cn=1.8 is assumed')
   30 format('        Kd implies that correction for eff. overburden',
     & ' pressure > 1.5 tons/sq.ft was applied. )
c
 1000 write(16,8)
c
      write(6,4)
      write(6,4)
      write(6,4)
      write(6,*) 'RELA: basic units are in LBS and FT'
      write(6,4)
      write(6,*) ' enter title of this run in 72 characters or less'
      write(6,4)
      read(5,2) title
      write(6,4)
      write(6,*) ' site description: enter no. of layers (<10)'
      write(6,4)
      read*, nlayer
      do  40 i=1,nlayer
      write(6,6) i
      write(6,4)
      read*, depth(i), den(i)
   40 continue
      th(1) = depth(1)
      do  60 i=2,nlayer
      th(i) = depth(i) - depth(i-1)
   60 continue
      write(6,4)
      write(6,*) ' enter expected depth of ground water during'
      write(6,*) ' the design earthquake, and ground water depth'
      write(6,*) ' when penetration test was performed -  7.0, 20.0'
      write(6,4)
      read*, zgw, zgwt
```

```
      write(6,4)
      write(6,*) ' enter equake mag. and max acc (g) -- 7.5, 0.25'
      write(6,4)
      read*, eqm, amax
      write(6,4)
      write(6,*) ' enter depth increment, dz (1.0 to 5. ft)'
      write(6,*) ' and relative density (0.4 for 40%), with a MINUS'
      write(6,*) ' sign if sand is silty'
      write(6,4)
      read*, dz, rden
      write(6,4)
      isilt = 0
      zlimit = 80.0
      if(depth(nlayer) .lt. 80.0) zlimit=depth(nlayer)
      if(rden .lt. 0.) isilt=1
      rden = abs(rden)
c
c        to check if gravelly deposit is being considered
c
      write(6,*) ' enter 0 if deposit is not gravelly'
      write(6,4)
      read*, igrav
      if(igrav .eq. 0) go to 70
      write(16,66)
   66 format(/' * * * correction for gravelly sands included * * *'/)
      write(6,*) ' enter multiple of time required for gravelly soil'
      write(6,*) ' to reach peak pore pressure when compared to'
      write(6,*) ' that for sand -- 4.1'
      write(6,4)
      read*, gfac
   70 continue
c
      call getfac(eqm,fac,igrav,gfac)
c
c        to establish reference stress-ratio vs n1 curve
c
      do  80 i=1,9
      yt(i) = ut(i)*fac
   80 continue
c
      z = zgw
      if(z .le. 0.) z=1.0
      dbc = 0.0
      if(isilt .eq. 1) dbc=7.5
      ic = 0
c
  100 ic = ic+1
c
      xshw = 1.0
      if(z .lt. 10.0) xshw=0.75
      xcpt = 4.5/xshw
      if(isilt .eq. 1) xcpt=4.0/xshw
      icn = 0
c
```

```
      call stress(z,sum1,sum2,s3,s4)
c
c
c        to determine stress reduction factor rd & ave stress-ratio
c
      j = 1
      do 220 loop=1,8
      j = j+1
      if(dref(j) .gt. z) go to 240
  220 continue
  240 fac1 = rd(j-1) + (z-dref(j-1))*(rd(j)-rd(j-1))/(dref(j)-dref(j-1))
      atau = 0.65*fac1*amax*sum2
      taur = atau/sum1
c
c        to determine modified penetration resistance, N1
c
      kdpt = 0
      facdpt = 1.0
      if(sum1 .gt. 3000.) kdpt=1
c
c        correction for overburden pressure .gt. 1.5 tsf according
c          to Seed, 1983, as listed in the documentation of RELA
c
      if(kdpt .ne. 0) facdpt=1.07-3.348*0.01*0.001*sum1
      taud = taur/facdpt
c
      if(taud .le. yt(9)) go to 320
      i=8
      go to 380
  320 if(taud .ge. yt(1)) go to 340
      i=1
      go to 380
  340 do 360 i=1,8
      if(taud .le. yt(i+1)) go to 380
  360 continue
  380 bcmod = xn(i)+(taud-yt(i))*(xn(i+1)-xn(i))/(yt(i+1)-yt(i))
c
c          to determine Cn from S3, effective stress during testing
c
      ysig = s3/1000.0
      if(rden .ge. 0.60) go to 480
      do 460 i=1,16
      xf(i) = cn4(i)
      yf(i) = sv4(i)
  460 continue
      go to 500
  480 do 490 i=1,16
      xf(i) = cn8(i)
      yf(i) = sv8(i)
  490 continue
  500 continue
      if(ysig .gt. yf(1)) go to 520
      icn = 1
      cn = 1.8
```

```
      go to 580
  520 continue
      j = 1
      do 540 loop=1,15
      j = j+1
      if(yf(j) .gt. ysig) go to 560
  540 continue
  560 cn = xf(j-1) + (xf(j)-xf(j-1))*(ysig-yf(j-1))/(yf(j)-yf(j-1))
  580 continue
      bctmp = (bcmod-dbc)/cn
      bc = bctmp/xshw
      cpt = bctmp*xcpt
c
      resu(1,ic) = z
      resu(2,ic) = sum1
      resu(3,ic) = sum2
      resu(4,ic) = s3
      resu(5,ic) = s4
      resu(6,ic) = taur
      resu(7,ic) = bcmod
      resu(8,ic) = bc
      resu(9,ic) = cpt
      resu(10,ic) = rmk(1)
      resu(11,ic) = rmk(1)
      if(isilt .eq. 1) resu(10,ic)=rmk(3)
      if(icn .eq.1) resu(11,ic)=rmk(4)
      if(kdpt .ne. 0) resu(11,ic)=rmk(5)
c
      if(bc .gt. 80.0) go to 800
      z = z+dz
      if(z .lt. zlimit) go to 100
c
c         save results on designated file for printer output
c
  800 continue
      write(16,2) title
      write(16,16) nlayer
      write(16,18) ((i,depth(i),den(i)),i=1,nlayer)
      write(16,26) rden
      write(16,20) eqm, amax, zgw, zgwt
      write(16,22)
      write(16,23)
      do 850 i=1,ic
      write(16,24) i,(resu(j,i),j=1,11)
  850 continue
      write(16,28)
      write(16,30)
c
c          check to see if new set of computation is needed
c
      write(6,4)
      write(6,*) ' enter integer>0 for a new set of computation'
      write(6,4)
      read*, icont
```

```
      if(icont .gt. 0) go to 1000
c
      stop
      end



      subroutine getfac(eqm,fac,igrav,gfac)
c
c        subroutine to compute scaling factor, fac, for a
c        given earthquake magnitude, eqm, to establish the
c        reference liquefaction potential curve --
c        stress ratio versus modified penetration blow count
c
      dimension sy(6),qx(6),cy(6)
c
c        digitized values of curve in fig. 56, ref. 1
c
      data sy/1.6,1.32,1.13,1.0,0.89,0.80/
      data qx/5.25,6.0,6.75,7.5,8.5,9.9/
      data cy/3.0,6.0,10.0,15.0,26.0,100.0/
c
      do 100 i=1,4
      if(eqm .le. qx(i+1)) go to 120
  100 continue
  120 cyn=cy(i)+(eqm-qx(i))*(cy(i+1)-cy(i))/(qx(i+1)-qx(i))
      if(igrav .ne. 0) cyn=cyn/gfac
      do 140 i=1,4
      if(cyn .le. cy(i+1)) go to 160
  140 continue
  160 continue
      delx = cyn/cy(i)
      dx = cy(i+1)/cy(i)
      fac = sy(i)+(sy(i+1)-sy(i))*alog(delx)/alog(dx)
      return
      end



      subroutine stress(z,s1,s2,s3,s4)
      common /blka/den(9),th(9),depth(9),nlayer,zg,zgwt
c
      iseq = 1
      zgw = zg
c
  100 continue
      if(iseq .eq. 2) zgw=zgwt
      sum1 = 0.0
      sum2 = 0.0
      if(z .gt. zgw) go to 220
      j = 0
      do 120 loop=1,nlayer
```

```
      j = j+1
      if(depth(j) .ge. z) go to 140
      sum1 = sum1+th(j)*den(j)
      sum2 = sum1
  120 continue
  140 if(j .gt. 1) go to 160
      sum1 = z*den(j)
      sum2 = sum1
      go to 400
  160 sum1 = sum1 + (z-depth(j-1))*den(j)
      sum2 = sum1
      go to 400
  220 continue
      j = 0
      do 240 loop=1,nlayer
      j = j+1
      if(depth(j) .ge. zgw) go to 250
      sum1 = sum1 + th(j)*den(j)
      sum2 = sum2 + th(j)*den(j)
  240 continue
  250 continue
      idry = j
      if(idry .gt. 1) go to 280
      if(z .gt. depth(1)) go to 260
c
c         z, zgw both in layer 1
      sum1 = zgw*den(1) + (z-zgw)*(den(1)-62.4)
      sum2 = z*den(1)
      go to 400
c
  260 sum1 = zgw*den(1) + (depth(1)-zgw)*(den(1)-62.4)
      sum2 = depth(1)*den(1)
      go to 320
  280 if(z .gt. depth(idry)) go to 300
      sum1 = sum1 + (zgw-depth(idry-1))*den(idry)
     &            + (z-zgw)*(den(idry)-62.4)
      sum2 = sum2 + (z-depth(idry-1))*den(idry)
      go to 400
  300 sum1 = sum1 + (zgw-depth(idry-1))*den(idry)
     &            + (depth(idry)-zgw)*(den(idry)-62.4)
      sum2 = sum2 + th(idry)*den(idry)
  320 continue
      do 340 loop=idry,nlayer
      j = j+1
      if(depth(j) .gt. z) go to 360
      sum1 = sum1 + th(j)*(den(j)-62.4)
      sum2 = sum2 + th(j)*den(j)
  340 continue
  360 sum1 = sum1 + (z-depth(j-1))*(den(j)-62.4)
      sum2 = sum2 + (z-depth(j-1))*den(j)
  400 continue
      if(iseq .eq. 2) go to 500
      s1 = sum1
      s2 = sum2
```

```
      iseq = 2
      go to 100
  500 s3 = sum1
      s4 = sum2
      return
      end
```